
A BSTRACT: The tautological ring of the moduli space of stable curves has been studied extensively in
the last decades. We present a SageMath implementation of many core features of this ring. This includes
lists of generators and their products, intersection numbers and verification of tautological relations.
Maps between tautological rings induced by functoriality, that is pushforwards and pullbacks under
gluing and forgetful maps, are implemented. Furthermore, many interesting cycle classes, such as the
double ramification cycles, strata of k-differentials and hyperelliptic or bielliptic cycles are available. We
show how to apply the package, including concrete example computations.

1. I NTRODUCTION . A crucial tool in the study of the singular cohomology of the moduli space Mg,n
of stable curves is the tautological ring
RH∗ (Mg,n ) ⊂ H ∗ (Mg,n ) = H ∗ (Mg,n , Q).
It is a Q-subalgebra of the singular cohomology of Mg,n with an explicit, finite set of generators
(indexed by decorated graphs [0, α]) admitting combinatorial descriptions of operations like cup products
and intersection numbers. For a detailed introduction to the tautological ring, see, e.g., [Faber and
Pandharipande 2000; Arbarello et al. 2011; Pandharipande 2018].
Since computations with the generators [0, α] quickly become untractable by hand, it is natural to
implement them in a computer program. With admcycles we present such an implementation using
the open source mathematical software [SageMath]. It is based on an earlier implementation by Aaron
Pixton. It features intersection products and numbers between the classes [0, α] and verification of linear
relations between these generators using the known generalized Faber–Zagier relations [Pixton 2012;
Pandharipande et al. 2015; Janda 2017]. For the gluing and forgetful morphisms between (products
of) the moduli spaces Mg,n it implements pullbacks and pushforwards of the generators [0, α] of the
tautological ring.
Many geometric constructions of cohomology classes on Mg,n (such as the Chern classes λd of the
Hodge bundle E over Mg,n ) give classes contained in the tautological ring and can thus be written as
linear combinations of classes [0, α]. For many examples of such classes, the package admcycles
MSC2010: 14H10, 97N80.

implements known formulas or algorithms to calculate them and thus allows further computations, such
as intersections or comparisons to other cohomology classes. In particular, admcycles contains
•
•

•

•

a formula for double ramification cycles DRg (A) from [Janda et al. 2017] ,
k

a conjectural formula for the strata Hg (m) of k-differentials from [Farkas and Pandharipande 2018;
Schmitt 2018],
(generalized) lambda classes, the Chern classes of derived pushforwards R • π∗ O(D) of divisors D
on the universal curve π : Cg,n → Mg,n , as discussed in [Pagani et al. 2020],
admissible cover cycles,1 such as the fundamental classes of loci of hyperelliptic or bielliptic curves
with marked ramification points, as discussed in [Schmitt and van Zelm 2020]

Instead of discussing the details of the algorithms in admcycles, this document serves as a user manual
for the package, with an emphasis on concrete example computations. These computations are also
available in an interactive online format on CoCalc (without need for registration) here.
One way to explore admcycles is to go through these examples and refer back to the text below
for additional explanations and background. While the code in the examples is mostly self-explanatory,
some basic familiarity with SageMath and the Python programming language (e.g., as explained in the
official SageMath tutorial) is helpful.
Applications of admcycles. By now the package admcycles has been used in a variety of contexts.
Its original purpose was computing new examples of admissible cover cycles in [Schmitt and van Zelm
2020], e.g., computing the class of the hyperelliptic locus in M5 and M6 and the locus of bielliptic cycles
in M4 . It was also used to verify results about Hodge integrals on bielliptic cycles in [Pandharipande
and Tseng 2019] and on loci of cyclic triple covers of rational curves in [Owens and Somerstep 2019].
Buryak and Rossi [2021] used admcycles to explore formulas for intersection numbers involving
double ramification cycles and lambda classes. The implementation of generalized lambda classes led
to the discovery of previously missing terms in the computations of [Pagani et al. 2020] when doing
comparisons with double ramification cycles. The package was also used in [Chen et al. 2019] to verify
computations of Masur–Veech volumes in terms of intersection numbers on Mg,n . It was used to check
a new recursion for intersection numbers of ψ-classes presented in [Grosse et al. 2019] and formulas for
double Hurwitz numbers in terms of intersection numbers in [Borot et al. 2020] and [Do and Lewański
2020]. In [Castorena and Gendron 2020], which computes a fundamental class of a stratum of meromorphic differentials in genus 3, some errors have been found and corrected after comparing the result with
the output of admcycles. More recently, in [Bae and Schmitt 2020] some code based on admcycles was
used to compute ranks of Chow groups of moduli stacks M0,n of prestable curves. Molcho et al. [2021]
applied the package to verify the completeness of the generalized Faber–Zagier relations in two new cases
on M4,1 and M5,1 and used this to show that for g ≥ 7 the class λg is not contained in the subring of the
1 Computing these cycles was the original purpose of admcycles, hence the name of the package.

cohomology of Mg generated by classes of cohomological degree at most 4. Very recently, Canning and
Larson [2021] used admcycles for computing the rational Chow rings of the spaces Mg for g = 7, 8, 9.
Other implementations. Apart from admcycles (and the code of Pixton on which it is based) there
have been several other implementations of the tautological ring, starting with Faber’s program [1999] for
computing intersection numbers of divisors and Chern classes of the Hodge bundle. Yang [2008] presents
a program computing intersection pairings of tautological classes on various open subsets of Mg,n . The
package [mgn] by Johnson implements general intersections of the [0, α] and also verification of linear
relations between these generators against the known generalized Faber–Zagier relations.
Based on admcycles there is the new SageMath-package diffstrata (included in admcycles since
version 1.1) by Costantini, Möller and Zachhuber. It implements the tautological ring and intersection
products on the smooth compactification of the strata of differentials presented in [Bainbridge et al.
2019a]. Computations with diffstrata are used in [Costantini et al. 2020a] to evaluate formulas
for Euler characteristics of strata of differentials in examples. Similar to the present paper, a detailed
description of the package diffstrata is given in [Costantini et al. 2020b].
1.1. Conventions. Let Mg,n be the moduli space of stable curves and π : Mg,n+1 → Mg,n be the
forgetful morphism of the marking n + 1, which can be seen as the universal curve over Mg,n . Let
σi : Mg,n → Mg,n+1 be the section of π corresponding to the i-th marked point (i = 1, . . . , n). For ωπ
the relative dualizing line bundle of π on the space Mg,n+1 and i = 1, . . . , n we define the ψ-class
ψi = c1 (σi∗ ωπ ) ∈ H 2 (Mg,n ).
For a = 0, 1, 2, . . . we define the (Arbarello–Cornalba) κ-class
κa = π∗ ((ψn+1 )a+1 ) ∈ H 2a (Mg,n ).
Finally, given a stable graph 0 of genus g with n legs, let
Y
ξ0 : M0 =
Mg(v),n(v) → Mg,n
v∈V (0)

be the gluing map associated to 0. For a class α ∈ H ∗ (M0 ) given as a product of κ and ψ-classes on
the factors Mg(v),n(v) , define
[0, α] = (ξ0 )∗ α ∈ H ∗ (Mg,n ).
Such decorated boundary strata form a generating set (as a Q-vector space) of the tautological ring
RH∗ (Mg,n ).
Note: The degree of the gluing map ξ0 to its image is given by the size |Aut(0)| of the automorphism
group of 0. Therefore many authors prefer to define [0, α] as 1/|Aut(0)| · (ξ0 )∗ α (so that [0, 1] equals
the class of the boundary stratum of Mg,n associated to 0). However, throughout this paper and in the
package admcycles, we take the convention of not dividing by the size |Aut(0)| of the automorphism
group of 0.

2. G ETTING STARTED . The admcycles package works on top of SageMath which is an open source
software for mathematical computations. We describe how to install SageMath and admcycles on a
computer and how to use the available online services.
2.1. admcycles in the cloud. The simplest way to play with admcycles without installing anything
beyond a web browser is to use one of [SageMathCell] or the website [CoCalc]. The former provides a
basic interface to SageMath. The latter requires registration and allows one to create worksheets that can
easily be saved and shared. As mentioned before, it is possible to explore the computations presented
below on share.cocalc.com without the need to register.
2.2. Obtaining SageMath. SageMath is available on most operating systems. Depending on the situation one can find it in the list of softwares available from the package manager of the operating system.
Alternatively, there are binaries available from the SageMath website . Lastly, one can compile it from
the source code. More information on the installation process can be found here.
2.3. Installation of the admcycles package. The package admcycles is available from the Python
Package Index (PyPI) where detailed installation instructions are available for a range of systems. Note
that the best performance (in particular for functions like DR_cycle) is obtained using version 9.0 of
SageMath or newer.
The package admcycles is being developed on GitLab where one can find the latest development
version and a link to report bugs. This is also the place to look at to suggest features or improvements.
2.4. First step with admcycles. Once successfully installed, to use admcycles one should start a
SageMath-session and type
sage: from admcycles import *

In the sample code, we reproduce the behavior of the SageMath console that provides the sage: prompt
on each input line. When using the online SageMathCell or a Jupyter worksheet, there is no need to write
sage:. In all our examples, this sage: prompt allows one to distinguish between the input (command)
and the output (result). All other examples below assume that the line
from admycles import *
has been executed before.
In addition to this manual, the package has an internal documentation with more information concerning the various functions. To access additional information about some function or object foo, type foo?
during the SageMath session; e.g.,
sage: TautologicalRing?

3. TAUTOLOGICAL RING AND CLASSES . The main objects in admcycles to manipulate tautological
classes are TautologicalRing and TautologicalClass.

3.1. Creating tautological rings. A convenient way to start a computation in the tautological ring of
Mg,n is to construct the appropriate ring itself by calling the function TautologicalRing(g, n).
sage: R = TautologicalRing(1, 1); R
TautologicalRing(g=1, n=1, moduli=’st’) over Rational Field

As we explain in Section 3.2, the object R above then allows easy access to many of the standard tautological classes on Mg,n . As an example, we show how to compute the integral
Z
1
ψ1 = 24
M1,1

using the ring R we created above (see Section 3.3 for more details):
sage: R.psi(1).evaluate()
1/24

Instead of working with the tautological ring of all of Mg,n , it is also possible to work on open subsets
of the moduli space, such as the locus of compact type curves. This can be specified with the parameter
moduli:
sage: Rct = TautologicalRing(3, 1, moduli=’ct’)

The available moduli types are:
•

’st’: all stable curves (default).

•

’tl’: treelike curves (all cycles in the stable graph have length 1).

•

’ct’: compact type (stable graph is a tree).

•

’rt’: rational tails (there exists a vertex of genus g).

•

’sm’: smooth curves.

As an example of how this affects the behavior of the tautological ring, we can compute the so-called
socle degree, i.e., the highest nonvanishing (complex) degree of the tautological ring of the corresponding
subset of Mg,n .
sage: Rst = TautologicalRing(3, 1, moduli=’st’)
sage: Rst.socle_degree()
7
sage: Rsm = TautologicalRing(3, 1, moduli=’sm’)
sage: Rsm.socle_degree()
2

We will see in more detail in Section 3.4 how specifying the moduli affects computations.
3.2. Creating tautological classes. Each tautological class in admcycles has type TautologicalClass.
We list in this section the different ways to enter tautological classes in the program. Depending on the
example, some are more convenient than others.
As explained in Section 3.1 all computations happen in a given tautological ring (with a fixed base
ring and fixed moduli). Once a tautological ring R for Mg,n has been created as explained in Section 3.1,

the fundamental class, boundary divisors as well as ψ, κ and λ-classes are predefined methods of the
ring R.
•
•

R.fundamental_class() returns the fundamental class of Mg,n .
R.separable_boundary_divisor(h,A) gives the pushforward ξ∗ [M0 ] of the boundary gluing
map
ξ : M0 = Mh,A∪{ p} × Mg−h,({1,...,n}\A)∪{ p0 } → Mg,n ,
where A can be a list, set or tuple2 of numbers from 1 to n.

•

R.irreducible_boundary_divisor() gives the pushforward (ξ 0 )∗ [Mg−1,n+2 ] of the boundary
gluing map
ξ 0 : Mg−1,n+2 → Mg,n
identifying the last two markings to a node. Note that, since ξ 0 has degree 2 onto its image, this
gives twice the fundamental class of the boundary divisor of irreducible nodal curves.

•

R.psi(i) gives the ψ-class ψi of marking i on Mg,n .

•

R.kappa(a) gives the (Arbarello–Cornalba) κ-class κa on Mg,n .

•

R.lambdaclass(d) gives the class λd on Mg,n , defined as the d-th Chern class λd = cd (E) of the
Hodge bundle E, the vector bundle on Mg,n with fiber H 0 (C, ωC ) over the point (C, p1 , . . . , pn ) ∈
Mg,n .

These tautological classes can be combined in the usual way by operations +, -, * and raising to an
integral power ^.
sage: R1 = TautologicalRing(3, 4)
sage: t1 = 3*R1.separable_boundary_divisor(1,(1,2)) - R1.psi(4)^2
sage: R2 = TautologicalRing(2, 1)
sage: t2 = -1/3*R2.irreducible_boundary_divisor() * R2.lambdaclass(1)

For user convenience, alternative functions are available to create the basic tautological classes (over the
rationals and for the full moduli of stable curves), without having to create the tautological ring before.
Each of these functions require extra arguments g and n to specify the genus and the number of marked
points.
•

fundclass(g, n)

•

sepbdiv(g1, A, g, n)

•

irrbdiv(g, n)

•

psiclass(i, g, n)

•

kappaclass(a, g, n)

•

lambdaclass(d, g, n)

2 Be careful that tuples of length 1 must be entered as (a,) in Python, instead of (a).

sage: tt1 = 3 * sepbdiv(1, (1,2), 3, 4) - psiclass(4, 3, 4)^2
sage: t1 == tt1
True
sage: tt2 = -1/3*irrbdiv(2, 1) * lambdaclass(1, 2, 1)
sage: t2 == tt2
True

To enter more complicated classes coming from decorated boundary strata, it is often convenient to
first list all such decorated strata forming the generating set of RH2r (Mg,n ) in a specified degree r using
R.list_generators(r) and then select the desired ones from the list (see below for an explanation
of the notation). As a shortcut one can also directly use the function tautgens(g,n,r) to produce this
list without having to create the ring R before.
sage: R = TautologicalRing(2, 0)
sage: R.list_generators(2)
[0] : Graph :
[2] [[]] []
Polynomial : (kappa_2)_0
[1] : Graph :
[2] [[]] []
Polynomial : (kappa_1^2)_0
[2] : Graph :
[1, 1] [[2], [3]] [(2, 3)]
Polynomial : (kappa_1)_0
[3] : Graph :
[1, 1] [[2], [3]] [(2, 3)]
Polynomial : psi_2
[4] : Graph :
[1] [[2, 3]] [(2, 3)]
Polynomial : (kappa_1)_0
[5] : Graph :
[1] [[2, 3]] [(2, 3)]
Polynomial : psi_2
[6] : Graph :
[0, 1] [[3, 4, 5], [6]] [(3, 4), (5, 6)]
Polynomial : 1
[7] : Graph :
[0] [[3, 4, 5, 6]] [(3, 4), (5, 6)]
Polynomial : 1

The list itself is created by R.generators(r), from which one can then select the classes:
sage: L = R.generators(2)
sage: t3 = 2*L[3]+L[4]
sage: t3
Graph :
[1] [[2, 3]] [(2, 3)]
Polynomial : (kappa_1)_0
Graph :
[1, 1] [[2], [3]] [(2, 3)]
Polynomial : 2*psi_2

The output above should be interpreted as follows: each TautologicalClass consists of a sum of
decorated boundary strata (represented by data type decstratum), which consist of a graph (datatype
StableGraph) and a polynomial in κ and ψ-classes (datatype KappaPsiPolynomial).
To explain the notation above, let us look at the example of generator L[3].
Graph :
[1, 1] [[2], [3]] [(2, 3)]
Polynomial : 1*psi_2^1

Its stable graph is represented by three lists.
(1) The first list [1, 1] are the genera of the vertices, so there are two vertices, both of genus 1. Note
that vertices are numbered by 0, 1, 2, . . . , so in the above case, the vertices are numbers 0 and 1.

(2) The second list gives the legs (that is markings or half-edges) attached to the vertices, so vertex 0
carries the half-edge 2 and vertex 1 the half-edge 3.
(3) The third list gives the edges, that is half-edge pairs that are connected; in the above case, the two
half-edges 2 and 3 form an edge, connecting the two vertices.
If we wanted to enter this StableGraph manually, we could use its constructor as follows:
sage: G = StableGraph([1,1],[[2],[3]],[(2,3)]); G
[1, 1] [[2], [3]] [(2, 3)]

The polynomial in κ and ψ is 1*psi_2^1 in this case, so the half-edge 2 on the first vertex carries a
ψ-class. For the generator L[4] the polynomial looks like 1*(kappa_1^1)_0, meaning that vertex 0
carries a class κ11 = κ1 .
Finally, it is possible to manually enter tautological classes by constructing a stable graph gamma and
calling the main constructor R(gamma,kappa,psi) of the tautological ring.
sage: R = TautologicalRing(3,2)
sage: g = StableGraph([2,0], [[1,3],[2,4,5,6]], [(3,4),(5,6)])
sage: R(g, kappa=[[],[1]], psi={1:2})
Graph :
[2, 0] [[1, 3], [2, 4, 5, 6]] [(3, 4), (5, 6)]
Polynomial : (kappa_1)_1*psi_1^2
sage: R(g, kappa=[[1,1],[]])
Graph :
[2, 0] [[1, 3], [2, 4, 5, 6]] [(3, 4), (5, 6)]
Polynomial : (kappa_1*kappa_2)_0

In the above call, the arguments kappa and psi are both optionals and specify the κ and ψ decorations
on the stable graph gamma. We refer to the documentation of admcycles for more details.
3.3. Basic operations. Apart from the usual arithmetic operations, we can take forgetful pushforwards
and pullbacks of tautological classes and also compute the degree of tautological zero-cycles. In particular, we can compute intersection numbers. Below, for the forgetful map π : M1,3 → M1,2 forgetting
the marking 3 we verify the relations
π∗ ψ32 = κ1

and

π ∗ ψ2 = ψ2 − D0,{2,3} ,

where D0,{2,3} is the class of the boundary divisor in M1,3 where generically the curve splits into two
components of genera 0, 1 connected at a node with the component of genus 0 carrying markings 2, 3.
sage: s1 = TautologicalRing(1, 3).psi(3)^2
sage: s1.forgetful_pushforward([3])
Graph :
[1] [[1, 2]] []
Polynomial : (kappa_1)_0
sage: s2 = TautologicalRing(1, 2).psi(2)
sage: s2.forgetful_pullback([3])
Graph :
[1] [[1, 2, 3]] []
Polynomial : psi_2
Graph :
[1, 0] [[1, 4], [2, 3, 5]] [(4, 5)]
Polynomial : -1

Using the method evaluate of TautologicalClass, we also compute intersection numbers of ψclasses on Mg,n , the so-called correlators or descendent integrals. Here, given numbers k1 , . . . , kn
summing to 3g − 3 + n one can define these correlators hτk1 · · · τkn ig,n as
Z
hτk1 · · · τkn ig,n =
ψ1k1 · · · ψnkn .
(1)
Mg,n

Below we compute the intersection number
hτ0 τ1 τ2 i1,3 =

Z
M1,3

ψ10 ψ2 ψ32 =

1
12

and check that it agrees with the prediction hτ0 τ1 τ2 i1,3 = hτ0 τ2 i1,2 + hτ12 i1,2 by the string equation.
sage: R1 = TautologicalRing(1, 3)
sage: s3 = R1.psi(2) * R1.psi(3)^2
sage: s3.evaluate()
1/12
sage: R2 = TautologicalRing(1, 2)
sage: s4 = R2.psi(2)^2 + R2.psi(1) * R2.psi(2)
sage: s4.evaluate()
1/12

Instead of multiplying ψ-classes and evaluating by hand, we can also use the function psi_correlator,
which takes as input the numbers k1 , . . . , kn and outputs the correlator (1).
sage: psi_correlator(0,1,2)
1/12

Note that in the current version of admcycles, the list of tautological generators [0i , αi ] in a tautological
class is not automatically simplified by combining equivalent terms (since in general this requires testing
graph isomorphisms between the 0i ). When performing arithmetic operations with complicated tautological classes, this simplification can be manually triggered using the function simplify, as demonstrated
below. For this toy example, we create two different but isomorphic stable graphs, convert them to
tautological classes and form their sum s. After applying the method simplify they are recognized as
equal, so that we obtain a shorter sum.
sage: gamma1 = StableGraph([1,2],[[3],[4]],[(3,4)]).to_tautological_class()
sage: gamma2 = StableGraph([2,1],[[5],[6]],[(5,6)]).to_tautological_class()
sage: s = gamma1 + gamma2; s
Graph :
[1, 2] [[3], [4]] [(3, 4)]
Polynomial : 1
Graph :
[2, 1] [[5], [6]] [(5, 6)]
Polynomial : 1
sage: s_simple = s.simplify(); s_simple
Graph :
[1, 2] [[2], [3]] [(2, 3)]
Polynomial : 2

In a future version of admcycles (after improving our algorithms for graph isomorphisms), we plan to
automate this process.

3.4. A basis of the tautological ring and tautological relations. One can compute, using the function
generating_indices(g,n,r), the indices (for the list tautgens(g,n,r)) of a basis of RH2r (Mg,n ),
assuming that the generalized Faber–Zagier relations (see [Pixton 2012; Pandharipande et al. 2015;
Janda 2017]) between the additive generators [0, α] give a complete set of relations between them. For
many concrete examples of (g, n, r ), this conjecture can be checked using admcycles via the function
FZ_conjecture_holds(g,n,r) (see [Molcho et al. 2021, Appendix B] and the documentation of the
function for more details). For the computation we show below, let us verify that the generalized Faber–
Zagier relations for RH2·2 (M2,0 ) are complete:
sage: FZ_conjecture_holds(2,0,2)
True

If the relations are complete as discussed above, Tautvecttobasis converts a vector with respect to the
whole generating set into a vector in this basis. The function TautologicalClass.basis_vector(r)
converts a TautologicalClass into such a vector.
Continuing the example from Section 3.2 we see:
sage: generating_indices(2,0,2)
[0, 1]
sage: t3.basis_vector(2)
(-48, 22)

This means that the generators L[0] and L[1] form a basis of RH4 (M2 ) and the TautologicalClass
t3=2*L[3]+L[4] is equivalent to -48*L[0]+22*L[1].
It is also possible to directly verify tautological relations using the built-in function is_zero of
TautologicalClass. It checks if the tautological class is contained in the ideal generated by the 3spin relations [Pandharipande et al. 2015] (what we call the generalized Faber–Zagier relations above).
Below we verify the known relation κ = ψ − δ0 ∈ R 1 (M1,n ) for n = 4. Here ψ is the sum of all ψi
and δ0 is the sum of all separating boundary divisors, i.e., those having a genus 0 component. For this,
we list all stable graphs with one edge via list_strata(g,n,1). We exclude the graph gamma with a
self-loop by requiring that the number of vertices gamma.numvert() is at least 2. Then we can convert
these graphs bd to tautological classes by using to_tautological_class.
sage: R = TautologicalRing(1, 4)
sage: bgraphs = [bd for bd in list_strata(1,4,1) if bd.num_verts() > 1]
sage: del0 = sum(bd.to_tautological_class() for bd in bgraphs)
sage: psisum = sum(R.psi(i) for i in range(1,5))
sage: rel = R.kappa(1) - psisum + del0
sage: rel.is_zero()
True

As a shorthand for is_zero one can also simply compare to the integer 0 as follows:
sage: rel == 0
True

It is also possible to express tautological classes in a basis of the tautological ring of suitable open subsets
of Mg,n , e.g., to verify that some relation holds on the locus of compact type curves. This works with

the optional argument moduli of TautologicalRing that was described in Section 3.1. We recall that
moduli can be one of ’st’ (stable), ’tl’ (treelike), ’ct’ (compact type), ’rt’ (rational tails) or ’sm’
(smooth). The functions basis_vector and is_zero depend very much on the underlying moduli. For
instance, we can verify the relation
λ1 =

B2
2 κ1

=

1
12 κ1

∈ H 2 (Mg )

following from Mumford’s computation [1983] in the case g = 3:
sage: R = TautologicalRing(3, 0, moduli=’sm’)
sage: R.kappa(1).basis_vector()
(1)
sage: R.lambdaclass(1).basis_vector()
(1/12)

It is also possible to start with a class on a bigger moduli (e.g., the default locus ’st’ of all stable curves)
and check whether it vanishes on a smaller subset using the optional parameter moduli of the functions
is_zero or basis_vector:
sage: R = TautologicalRing(2, 0)
sage: u = R.lambdaclass(2)
sage: u.is_zero()
False
sage: u.is_zero(moduli=’ct’)
True
sage: u.basis_vector()
(-3/2, 1/2)
sage: u.basis_vector(moduli=’ct’)
()

The vanishing here was expected as on Mct
2,0 the tautological ring in degree 2 vanishes:
sage: R = TautologicalRing(2, 0, moduli=’ct’)
sage: R.socle_degree()
1

In practice, much of the time in some computations is spent on calculating generalized Faber–Zagier
relations between tautological cycles on Mg,n . However, once computed, the relations can be saved to
a file and reloaded in a later session using the functions save_FZrels() and load_FZrels(). Careful:
the function save_FZrels() creates (and overwrites previous version of) a file new_geninddb.pkl
which, depending on the previous computations, can be quite large.
3.5. Pulling back tautological classes to the boundary. Recall that for a stable graph 0 we have a gluing
map
m
Y
ξ0 : M0 =
Mg(vi ),n(vi ) → Mg,n
(2)
i=1

taking one stable curve for each of the vertices v1 , . . . , vm of 0 and gluing them together according to
the edges of 0. By [Graber and Pandharipande 2003, Appendix A], the pullback of a tautological class

100

Delecroix, Schmitt and van Zelm

:::: admcycles Sage package

under ξ0 is contained in the tensor product of the tautological rings of the factors Mg(vi ),n(vi ) above, and
this operation is implemented in admcycles.
Below we pull back a generator of RH4 (M4 ) to the boundary divisor with genus partition 4 = 2 + 2.
This produces an element of type prodtautclass, a tautological class on a product of moduli spaces,
in this case M2,1 × M2,1 . Two elements on the same product of spaces can be added and multiplied and
further operations like pushforwards under (partial) gluing maps are supported. More details are given
in the documentation of the class prodtautclass.
Below, we want to express the pullback to M2,1 × M2,1 in terms of a basis of H 2 (M2,1 × M2,1 )
obtained from the preferred bases of the factors H ∗ (M2,1 ) given by generating_indices. We can
either represent the result as a list of matrices (giving the coefficients in the tensor product bases) or as
a combined vector (using the option vecout=true).
sage: bdry=StableGraph([2,2],[[1],[2]],[(1,2)])
sage: generator=tautgens(4,0,2)[3]; generator
Graph :
[1, 3] [[2], [3]] [(2, 3)]
Polynomial : psi_3
sage: pullback=bdry.boundary_pullback(generator)
sage: pullback.totensorTautbasis(2)
[
[-3]
[ 1]
[0 0 0] [-3]
[0 0 0] [ 7]
[-3 1 -3 7 1], [0 0 0], [ 1]
]
sage: pullback.totensorTautbasis(2,vecout=true)
(-3, 1, -3, 7, 1, 0, 0, 0, 0, 0, 0, 0, 0, 0, -3, 1, -3, 7, 1)

3.6. Pushing forward classes from the boundary. The pushforward under the map ξ0 in (2) sends a
product of tautological classes on the factors Mg(vi ),n(vi ) to a tautological class of Mg,n . This operation
is implemented by the function boundary_pushforward of StableGraph.
That is to say, if Gamma is a StableGraph and if [c1, ..., cm] is a list whose i-th element ci is
a TautologicalClass on the i-th factor Mg(vi ),n(vi ) of M0 , then
Gamma.boundary_pushforward([c1, ..., cm])
is the pushforward of the product of the ci. Here, the markings for the class ci are supposed to go from
1 to n(vi ), where the j-th marking corresponds to leg number j on the i-th vertex of Gamma.
As an illustration, we verify that the package correctly computes the excess intersection formula
proved in [Graber and Pandharipande 2003] for the self-intersection of a boundary divisor in M3,3 .
sage: B=StableGraph([2,1],[[4,1,2],[3,5]],[(4,5)])
sage: Bclass = B.boundary_pushforward() # class of undecorated boundary divisor
sage: si1 = B.boundary_pushforward([fundclass(2,3),-psiclass(2,1,2)])
sage: si1
Graph :
[2, 1] [[4, 1, 2], [3, 5]] [(4, 5)]
Polynomial : -psi_5
sage: si2 = B.boundary_pushforward([-psiclass(1,2,3),fundclass(1,2)])

sage: si2
Graph :
[2, 1] [[4, 1, 2], [3, 5]] [(4, 5)]
Polynomial : -psi_4
sage: (Bclass*Bclass-si1-si2).is_zero()
True

Note that, e.g., for the term si2 we needed to hand the function the term -psiclass(1,2,3) in the
first vertex, since in the graph B the half-edge 4 is leg number 1 in the list of legs at the first vertex (and
we have (g(v1 ), n(v1 )) = (2, 3) for this vertex).
4. S PECIAL CYCLE CLASSES . Beyond the already mentioned standard tautological classes ψi , κa and
λd and boundaries from Section 3.2, admcycles provides more advanced constructions that we describe
now. The corresponding functions are summarized here:
TautologicalRing method
double_ramification_cycle
theta_class
differential_stratum
generalized_lambda
hyperelliptic_cycle
bielliptic_cycle

standalone function
DR_cycle
ThetaClass
Strataclass
generalized_lambda
Hyperell
Biell

manual section
Section 4.1
Section 4.1
Section 4.2
Section 4.3
Section 4.4
Section 4.4

A convenient way to find out about tautological class constructions is to use the tab completion feature
of SageMath. When you enter a part of a name and press the tab key (denoted <TAB> below) the program
will show you all available completions. It can be used to discover the names in the admcycles module.
sage: import admcycles
sage: admcycles.<TAB>
admcycles.Biell
admcycles.DR
admcycles.DR_cycle
admcycles.DR_cycle_old

admcycles.DR_phi
admcycles.DRpoly
admcycles.FZ_conjecture_holds
admcycles.GRRcomp

...
...
...
...

Similarly one can discover the methods of TautologicalRing starting with the letter d:
sage: R = TautologicalRing(2, 2)
sage: R.d<TAB>
R.differential_stratum
R.dump
R.dimension
R.dumps
R.double_ramification_cycle

4.1. Double ramification cycles. A particularly interesting family of cycles on Mg,n is given by the
double ramification cycles. Fixing g, n they are indexed by nonnegative integers k, d ≥ 0 and a tuple
A = (a1 , a2 , . . . , an ) of integers summing to k(2g − 2 + n).
The classical double ramification cycle (for k = 0, d = g)
DRg (A) ∈ H 2g (Mg,n )
has been defined as the pushforward of the virtual fundamental class of a space of maps to rubber P1

relative to 0, ∞ with tangency conditions at 0, ∞ specified by the vector A (see [Li and Ruan 2001; Li
2002; 2001; Graber and Vakil 2005]). In [Janda et al. 2017] it is shown that this cycle is tautological and
an explicit formula in terms of tautological generators is provided.
More precisely, for g, n, k, d and A with A a partition of k(2g − 2 + n), the paper constructs an explicit
tautological class
Pgd,r,k (A) ∈ Q[r ] ⊗Q RH2d (Mg,n )
with coefficients being polynomials in a formal variable r . We obtain a usual tautological class Pgd,k (A) ∈
RH2d (Mg,n ) by setting r = 0 in these polynomial coefficients. Then it is shown ([Janda et al. 2017,
Theorem 1]) that in the special case k = 0, d = g, this gives a formula for the double ramification cycle
DRg (A) = 2−g Pgg,k (A).
While this demonstrates that the cycle Pgd,k (A) is useful for k = 0, d = g, it has many interesting properties
for other values of k, d:
•

For k arbitrary and d = 1, the restriction of 2−1 Pg1,k (A) to the compact-type locus Mct
g,n gives the
pullback of the theta divisor on the universal Jacobian J over Mct
under
the
extension
of the
g,n
Abel–Jacobi section
Mg,n → J ,

(C, p1 , . . . , pn ) 7→

log
(ωC )⊗k

 X

n
ai pi ;
−
i=1

see [Hain 2013; Grushevsky and Zakharov 2014].
•

•

For k arbitrary and d = g, various geometric definitions of a double ramification cycle have been
g,k
put forward and an equality with 2−g Pg (A) was conjectured in [Farkas and Pandharipande 2018;
Schmitt 2018] (see [Holmes and Schmitt 2019, Section 1.6] for an overview of the various definitions). Recently, this conjecture was proven in [Bae et al. 2020] based on earlier results of [Holmes
and Schmitt 2019].
For k arbitrary and d > g, the class Pgd,k (A) vanishes by [Clader and Janda 2018].

In admcycles, the formula for Pgd,k (A) has been implemented. The function DR_cycle(g,A,d,k)
returns the cycle 2−d Pgd,k (A). The factor 2−d was chosen such that DR_cycle(g,A) indeed gives the
cycle DRg (A). With the option rpoly=True, it is even possible to compute the cycle 2−d Pgd,r,k (A)
whose coefficients are polynomials in the variable r .
As an application, we can verify the result from [Holmes et al. 2019] that DR cycles satisfy the
multiplicativity property
DRg (A) · DRg (B) = DRg (A) · DRg (A + B) ∈ H 4g (Mtlg,n )
on the locus Mtlg,n of treelike curves but not on the locus of all stable curves, in the example given in
[Holmes et al. 2019, Section 8].

sage: A=vector((2,4,-6)); B=vector((-3,-1,4))
sage: diff = DR_cycle(1,A)*DR_cycle(1,B)-DR_cycle(1,A)*DR_cycle(1,A+B)
sage: diff.is_zero(moduli=’tl’)
True
sage: diff.is_zero(moduli=’st’)
False

In fact, using that the cycle DRg (A) is polynomial in the entries of the vector A (i.e., a tautological class
with polynomial coefficients), we can check multiplicativity for all vectors A, B in the case g = 1, n = 3.
To gain access to the polynomial-valued DR cycle, we define a polynomial ring and call DR_cycle with
a vector A having as coefficients the generators of this ring:
sage: R.<a1,a2,a3,b1,b2,b3> = PolynomialRing(QQ,6)
sage: A = vector((a1,a2,a3)); B = vector((b1,b2,b3))
sage: diff = DR_cycle(1,A)*DR_cycle(1,B)-DR_cycle(1,A)*DR_cycle(1,A+B)
sage: diff.is_zero(moduli=’tl’)
True

As a second application, we can verify the formula from [Buryak and Rossi 2021, Theorem 2.1] for
intersection numbers of two DR cycles with λg on Mg,3 in the case g = 1:
sage: intersect = DR_cycle(1,A)*DR_cycle(1,B)*lambdaclass(1,1,3)
sage: f = intersect.evaluate(); factor(f)
(1/216) * (a2*b1 - a3*b1 - a1*b2 + a3*b2 + a1*b3 - a2*b3)^2
sage: g = f.subs({a3:-a1-a2,b3:-b1-b2}); factor(g)
(1/24) * (a2*b1 - a1*b2)^2

The formula of the cycle Pgd,r,k (A) in [Janda et al. 2017] is obtained as a simplification (modulo r ) of
a cycle
r 2d−2g+1 ∗ cd (−R ∗ π∗ L)

(3)

appearing in [Janda et al. 2017, Corollary 4, Proposition 5] (see there for the notation). The cycle (3) is
often called a Chiodo class and it is relevant for certain computations (see [Borot et al. 2020; Do and
Lewański 2020]). Since the latest version of admcycles, the cycle (3) can be obtained using the optional
parameters chiodo_coeff = True and r_coeff of DR_cycle, which evaluates the expression (3) at
the value r_coeff of r .
sage: g=2; A=(5,-1); d=2; k=1
sage: Chiodo = DR_cycle(g,A,d,k,chiodo_coeff=True,r_coeff=7)

As a special case of this formula, we can obtain the cycle class θg,n ∈ R ∗ (Mg,n ) described in [Norbury
2017], which is accessible via the function ThetaClass.
sage: T = ThetaClass(1,1)
sage: T == 3*psiclass(1,1,1)
True

codim m = 0 m = k · m0 for m0 ∈ Zn≥0 m 6= k · m0
k=0
0
0
k=1
0
g−1
k>1
0
g − 1 and g

for m0 ∈ Zn≥0
g
g
g

k

Table 1. Dimension theory of Hg (m). Note that for k > 1 and m = k · m0 with m0 ∈ Zn≥0 ,
1

k

the set Hg (m0 ) ⊂ Hg (m) is a union of components of codimension g − 1 in Mg,n , with
k

all other components of Hg (m) having pure codimension g.
4.2. Strata of k-differentials. Let g, n, k ≥ 0 with 2g − 2 + n > 0 and let m = (m 1 , . . . , m n ) ∈ Zn with
P
i m i = k(2g − 2). Consider the subset

X


n
⊗k
k
∼
Hg (m) = (C, p1 , . . . , pn ) ∈ Mg,n : ωC
m i pi = OC ⊂ Mg,n .
i=1
k

Denote by Hg (m) the closure of Hkg (m) inside Mg,n . Since the above equality of line bundles is equivalent to the existence of a meromorphic k-differential η on C with zeros and poles exactly at the points
k
pi with multiplicities m i , the subsets Hg (m) are called strata of k-differentials.
These strata are of interest in algebraic geometry, the theory of flat surfaces and Teichmüller dynamics
and have been studied intensely in the past. Elements appearing in the boundary have been classified
in [Bainbridge et al. 2018; 2019b] and a smooth, modular compactification has been constructed in
k
[Bainbridge et al. 2019a]. The dimension of Hg (m) depends on k, m as in Table 1 (see, e.g., [Farkas and
Pandharipande 2018; Schmitt 2018]).
For k ≥ 1, [Farkas and Pandharipande 2018; Schmitt 2018] present conjectural relations between the
k
fundamental classes [Hg (m)] and the formulas for the double ramification cycles proposed by Pixton (see
Section 4.1). The conjectures were recently proven in [Bae et al. 2020] based on results from [Holmes
and Schmitt 2019]. As explained in the papers, these conjectures can be used to recursively determine
all cycles
k

•

[Hg (m)] ∈ RH2g (Mg,n ) for k ≥ 1 and m 6= km0 for some m0 ∈ Zn≥0 ,

•

[Hg (m)] ∈ RH2g−2 (Mg,n ) for k = 1 and m ∈ Zn≥0 .

1

These recursive algorithms have been implemented in the function Strataclass(g,k,m), where as
above m is a tuple of n integers summing to k(2g − 2).
1
As a small application, we can check that the stratum class [H2 ((3, −1))] vanishes (the stratum is
empty since by the residue theorem there can be no meromorphic differential with a single, simple
1
pole). Also, the stratum H2 ((2)) exactly equals the class of the locus of genus 2 curves with a marked
Weierstrass point, which can be computed by the function Hyperell (see below for details).
sage: L=Strataclass(2,1,(3,-1)); L.is_zero()
True
sage: L=Strataclass(2,1,(2,)); (L-Hyperell(2,1)).is_zero()
True

4.3. Generalized lambda classes. Let π : Cg,n → Mg,n be the universal curve and assume n ≥ 1. Every
divisor of Cg,n , up to pullback of divisors on Mg,n , takes the form
D = l K̃ +

n
X
p=1

dpσp +

X

ah,S C h,S

h≤g,
1∈S⊂[n]

for some integers l, d p , ah,S . Here K̃ = c1 (ωπ ) is the first Chern class of the relative dualizing sheaf, σ p
is the class of the p-th section and
C h,S = ξ∗ [Mh,S∪{•} × Mg−h,[n]\S∪{?,x} ] ∈ CH1 (Mg,[n]∪{x} ) = CH1 (Cg,n ).
The fact that every divisor D on Cg,n can be written in this form up to pullbacks from Mg,n follows from
the identification Cg,n ∼
= Mg,n+1 and the computation of the Picard group of the moduli spaces of stable
curves due to Harer [1983] and Arbarello and Cornalba [1987]. In [Pagani et al. 2020] a formula is given
for the Chern character ch(R • π∗ O(D)). This Chern character can be computed up to degree dmax using
generalized_chern_hodge(l,d,a,dmax,g,n). It takes as input an integer l, a list d=[d1,...,dn]
of the integers di and a list of triples a=[[h1,S1,ahS1],...,[hn,Sn,ahSn]] where the ahSi are the
integers ah,S above (given in any order). It is enough to just include the triples [h,S,ahS] for which
ah,S is nonzero.
Using generalized_lambda(i,l,d,a,g,n) the Chern class ci (−R • π∗ O(D)) can be computed
directly. In particular when l = 1 and the d p and ah,S are zero, this equals the normal λ class.
sage: g=3;n=1
sage: l=1;d=[0];a=[]
sage: s=lambdaclass(2,g,n)
sage: t=generalized_lambda(2,l,d,a,g,n)
sage: (s-t).is_zero()
True

Pn
0 be an element
Let d1 , ..., dn be integers such that i=1
di is divisible by 2g − 2 and let φ ∈ Vg,n
0 defined in [Kass and Pagani 2019, Definition 3.2]. This φ is an assignment
of the stability space Vg,n
which given a stable curve (C, p1 , . . . , pn ) ∈ Mg,n associates a real number φ(C, p1 , . . . , pn )C 0 to every
irreducible component C 0 of C. These numbers must sum to zero as C 0 runs through the components of C;
they only depend on the stable graph of C and must be compatible with degenerations of curves. Given
this data, Kass and Pagani construct a compactification J g,n (φ) of the universal Jacobian over Mg,n .
Pn
Let now l = i=1
di /(2g − 2) and let ah,S be integers such that
X
X
D(φ) = l K̃ +
di σi +
ah,S (φ)C h,S
is φ-stable on the locus of stable curves with one node (for definitions, see [Kass and Pagani 2019] or
[Pagani et al. 2020]). For the shifted3 vector A = (d1 + l, . . . , dn + l), [Holmes et al. 2018] proves
3 This shift is due to the fact that the literature on double ramification cycles uses the “log-convention”, i.e., the entries of the
input sum to l(2g − 2 + n).

an equality
DRg (A)|U (φ) = cg (−R • π∗ O(D(φ)))|U (φ)

(4)

on the largest open locus U (φ) ⊂ Mg,n where the Abel–Jacobi section
sl,d (φ) : Mg,n 99K J g,n (φ), (C, p1 , . . . , pn ) 7→ ωC⊗l −

n
X

!
di pi

i=1

extends to a morphism. In particular, U (φ) always includes Mct
g,n and equals Mg,n if and only if l, d
are trivial or l(2g − 2) = 0 and d = [0, ..., ±1, ..., ∓1, ..., 0]. See [Pagani et al. 2020, Section 4.3] for
more details.
The function DR_phi(g,d) computes cg (−R • π∗ O(D(φ))). We can verify equality (4).
sage: g=2;d=[1,-1]
sage: (DR_cycle(g,d)-DR_phi(g,d)).is_zero()
True

We also see that equality does not always hold over all of Mg,n but it does hold over Mct
g,n .
sage: g=2;d=[2,-2]
sage: (DR_cycle(g,d)-DR_phi(g,d)).basis_vector()
(12, -4, 14, 7, -40, -10, -14, -12, 28, -4, 6, -1, 4, 0)
sage: (DR_cycle(g,d)-DR_phi(g,d)).basis_vector(moduli=’ct’)
(0, 0, 0, 0, 0)

4.4. Admissible cover cycles.
Hyperelliptic and bielliptic cycles. Before we go into details of how to specify general admissible cover
cycles, let us mention the important cases of hyperelliptic and bielliptic cycles.
Recall that a smooth curve C is called hyperelliptic if C admits a double cover C → P1 and is called
bielliptic if it admits a double cover C → E of some smooth genus 1 curve E. In both cases we have an
involution C → C that exchanges the two sheets of the cover. Given g, n, m ≥ 0 with n ≤ 2g + 2 and
2g − 2 + n + 2m > 0, we have the locus H g,n,2m ⊂ Mg,n+2m which is the closure of the locus of smooth
curves (C, p1 , . . . , pn , q1 , q10 , . . . , qm , qm0 ) such that C is hyperelliptic with p1 , . . . , pn fixed points of the
hyperelliptic involution and the pairs qi , qi0 being exchanged by this involution. An analogous definition
gives the locus B g,n,2m ⊂ Mg,n+2m as the closure of the set of bielliptic curves with n ≤ 2g − 2 fixed
points and m pairs of points forming orbits under the bielliptic involution.
Then the fundamental class of the (reduced) loci H g,n,2m and B g,n,2m can (in many cases) be computed
by the functions Hyperell(g,n,m) and Biell(g,n,m) of our program.
As an example, we compute the class [H 3 ] ∈ RH2 (M3 ) and verify that we obtain the known result
[H 3 ] = 9λ − δ0 − 3δ1 ,
where δ0 is the class of the divisor of irreducible nodal curves and δ1 is the divisor of curves with a
separating node between a genus 1 and a genus 2 component.

sage: H = Hyperell(3,0,0)
sage: H.basis_vector()
(3/4, -9/4, -1/8)
sage: R = TautologicalRing(3, 0)
sage: H2 = 9*R.lambdaclass(1)-(1/2)*R.irreducible_boundary_divisor()
-3*R.separable_boundary_divisor(1,())
sage: H2.basis_vector()
(3/4, -9/4, -1/8)

Here we need to divide irrbdiv() by two, the degree of the corresponding gluing map.
Creating and identifying general admissible cover cycles. Generalizing the case of hyperelliptic and
bielliptic cycles, we can consider loci of curves C admitting a cover C → D to a second curve D such
that the cover is Galois with respect to a fixed finite group G. Cycles defined via such covers were
studied in [Schmitt and van Zelm 2020]. In general, such an admissible cover cycle is specified by the
genus g of the curve C, the finite group G, and monodromy data (we refer the reader to [Schmitt and
van Zelm 2020, Section 1.3] for the precise definitions). Currently, intersections are only implemented
for cyclic groups. Below we will study bielliptic curves in genus 2, which are double covers of elliptic
curves branched over two points. As a first step we enter the monodromy data.
sage: G=PermutationGroup([(1,2)])
sage: list(G)
[(), (1,2)]
sage: H=HurData(G,[G[1],G[1]])

The function HurData takes the group G as the first argument and as the second a list of group
elements α ∈ G, each of which corresponds to the G-orbit of some marking p ∈ C. Here α is a generator
of the stabilizer of p under the group action G y C, which gives the monodromy around p. In other
words, the natural action of the stabilizer G p = hαi on a tangent vector v ∈ T p C is given by
α.v = exp(2πi/ord(h))v.
Thus in the example above, we have two markings, both with stabilizer generated by G[1]=(1,2)
which acts by multiplication by −1 on the tangent space.
To identify the admissible cover cycle (inside the moduli space Mg,n with n the total number of
marked points from the monodromy data) in terms of tautological classes, one can use the function
Hidentify. It pulls back the admissible cover cycle to all boundary divisors and (recursively) identifies
the pullback itself in terms of tautological classes. It compares this pullback to the pullback of a basis
of the tautological ring. Often this pullback map is injective in cohomology so that one can then write
the admissible cover cycle in terms of the basis using linear algebra. Sometimes, it is necessary to
additionally intersect with some monomials in κ and ψ-classes.
To apply Hidentify one gives the genus and the monodromy data as arguments. The standard output
format is an instance of the class TautologicalClass. For users familiar with Pixton’s implementation
of the tautological ring, there is the option vecout=true which returns instead a vector with respect to

the generating set of the tautological ring provided by this program.
sage: vbeta = Hidentify(2, H, vecout=true)
sage: vector(vbeta)
(517/4, -33, 11/4, 243/4, -125/4, 15/2, 41/4, 125, 99/4, -41, -1137/4, -285/4, 0, 0, 0, 0, 0,
0, -57/8, -3/8, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0)

This output means, specifically, that inside M2,2 the locus of bielliptic curves with the two points fixed by
the involution being the marked points is given (in the generating set gens=all_strata(2,3,(1,2))
produced by Pixton’s program) as 517/4 · gens[0] − 33 · gens[1] + . . . .
If we instead wanted to have bielliptic curves with two marked fixed-points of the involution and one
pair of markings that are exchanged by the involution, we would need to use the monodromy data
sage: H2=HurData(G,[G[1],G[1],G[0]])

in which case Hidentify(2,H2) would live inside RH8 (M2,4 ).
If we only want to remember a subset of the markings, we can use the optional parameter marking to
give this subset. For instance, the command Hidentify(2,H,markings=[]) would give the pushforward of Hidentify(2,H) in M2,2 to the space M2 under the forgetful morphism (see also Section 4.4).
Example: specifying and identifying [B 2 ] by hand. The locus B 2 ⊂ M2 of bielliptic curves is a divisor. A
bielliptic genus 2 curve is ramified over two points. In the following we use the methods of the previous
section to identify its cycle class.
Now when treating admissible cover cycles in general, our program a priori handles the cycle where all
possible ramification points are marked. In this case, this is the cycle [B 2,2,0 ] ∈ RH6 (M2,2 ) of bielliptic
curves C with the two ramification points p1 , p2 marked. By specifying markings=[] when calling
Hidentify, we tell it to remember none of the markings, in other words to push forward under the map
π : M2,2 → M2 forgetting the markings.
sage: G = PermutationGroup([(1,2)])
sage: H = HurData(G, [G[1],G[1]])
sage: B22 = Hidentify(2, H, markings=[])
sage: B22.basis_vector(1)
(30, -9)

We compare the result with the known formulas for [B 2 ]. For δ0 the class of the irreducible boundary of
M2 and δ1 the class of the boundary divisor with genus-splitting (1, 1), it is known that [B 2 ] = 23 δ0 + 6δ1
(see [Faber 1996]). If we want to enter this combination of δ0 and δ1 , we have to be careful about
conventions, though: the corresponding gluing maps ξ : M1,2 → M2 and ξ 0 : M1,1 × M1,1 → M2 both
have degree 2. This corresponds to the fact that the associated stable graphs both have an automorphism
group of order 2. Hence we have to divide by a factor of two and obtain
sage: B22_formula = 3/4*irrbdiv(2,0) + 3*sepbdiv(1,(),2,0)
sage: B22_formula.basis_vector(1)
(15/2, -9/4)

We see that up to a factor of 4 the two vectors (30, -9) and (15/2, -9/4) agree. Where does this
factor come from?
For this, recall that the cycle B22 above is equal to π∗ [B 2,2,0 ]. Since for the generic bielliptic curve C
there are two choices of orderings for marking p1 , p2 , this explains a factor of 2. On the other hand, the
hyperelliptic involution σ : C → C on C exchanges p1 and p2 . Thus σ ∈ Aut(C), but σ ∈
/ Aut(C, p1 , p2 ).
This missing automorphism factor explains another factor of 2 in the pushforward under π, so in fact
[B 2 ] = 41 π∗ [B 2,2,0 ].
Note that since the cycles of bielliptic loci are implemented via the function Biell, we could have
taken a shortcut above.
sage: B = Biell(2,0,0)
sage: B.basis_vector()
(15/2, -9/4)

As an application, we can check the Hurwitz–Hodge integral
Z
Z
λ2 λ0 =
λ2 =
[B 2,2,0 ]

π∗ [B 2,2,0 ]

1
48

predicted by [Pandharipande and Tseng 2019].
sage: (B22 * lambdaclass(2,2,0)).evaluate()
1/48

The corresponding integrals for g = 3, 4 have also been verified like this, but the amount of time and
memory needed grows drastically.
We can also check the Hurwitz–Hodge integral
Z
λ1 = 92
[H2,Z/3Z,((1,2,3)2 ,(1,3,2)2 ) ]

of λ1 against the locus of genus 2 curves admitting a cyclic triple cover of a genus 0 curve with two
points of ramification (1, 2, 3) ∈ Z/3Z and two points of ramification (1, 3, 2) ∈ Z/3Z, computed in
[Owens and Somerstep 2019, Section 5].
sage: G = PermutationGroup([(1,2,3)]); sorted(list(G))
[(), (1,2,3), (1,3,2)]
sage: H = HurData(G,[G[1],G[1],G[2],G[2]]) #n=2, m=2
sage: t = Hidentify(2,H,markings=[])
sage: (t*lambdaclass(1,2,0)).evaluate()
2/9

Note that while originally the cycle [H2,Z/3Z,((1,2,3)2 ,(1,3,2)2 ) ] lives in M2,4 , since we intersect with λ1
which is a pullback from M2 we can specify markings=[] above to compute the pushforward t of this
cycle to M2 before intersecting. This significantly reduces the necessary computation time.

